Solving Equations Involving the Catenary with MLAB

Daniel R. Kerner, Ph.D.
Civilized Software, Inc.
12109 Heritage Park Circle
Silver Spring MD 20906
Tel.: (301)-962-3711
email: kerner@civilized.com
URL: www.civilized.com

A 10 foot rope hangs in the form of a catenary with one end attached to a pier at a point 3 feet above the water's surface, and the other end attached at the same height to a boat. If the lowest point on the catenary just touches the water, how far is the boat from the pier? If that distance is fixed, does the minimum of the catenary submerge or clear the surface of the water as the tide ebbs? as the tide flows?



Figure 1: A rope fixed to a pier and a boat. The lowest point touches the water's surface.

It has been known since Leibnitz and Huygens' time, circa 1691, that a rope or chain tethered at each end hangs in the shape of a catenary curve. Reference 1 gives the mathematical form of a catenary as:

$$y(x) = \beta + \gamma \cosh(\frac{x}{\gamma} + \alpha).$$

α, β, and γ must be chosen so that the catenary passes through two designated endpoints and the total length of the catenary has the proper value. The length of the catenary with abscissae ranging from x_0 to x_1 is given by:

$$L(x_0, x_1) = \gamma\ [\sinh(\frac{x_1}{\gamma} + \alpha) - \sinh(\frac{x_0}{\gamma} + \alpha)].$$

Choosing a coordinate system that has an x-axis co-linear with the line segment that is parallel to the water surface and connects the endpoints of the rope, and a y-axis that is perpendicular to the x-axis and passes through the minimum of the catenary gives rise to 4 equations:

$$\begin{aligned}
0 &= \beta + \gamma\ \cosh(\alpha) \\
3 &= \beta + \gamma\ \cosh(\frac{\bar{x}}{2\gamma} + \alpha) \\
3 &= \beta + \gamma\ \cosh(-\frac{\bar{x}}{2\gamma} + \alpha) \\
10 &= \gamma\ [\sinh(\frac{\bar{x}}{2\gamma} + \alpha) - \sinh(-\frac{\bar{x}}{2\gamma} + \alpha)]
\end{aligned} \tag{1}$$

where $\bar{x}$ is the distance separating the pier and the boat.

The MLAB mathematical modeling program provides several ways of solving sets of equations such as this. Here we describe 2 methods: one using the `ROOT` operator and the other using the `MINIMIZE` operator.

To use the MLAB `ROOT` operator effectively, a little more human thinking is required to reduce the 4 equations in 4 unknowns to 1 equation in 1 unknown. With some algebraic manipulation of the four equations, one finds $\alpha = 0$, $\beta = -(\gamma + 3)$, $\bar{x} = 2\gamma \sinh^{-1}(\frac{5}{\gamma})$, and that γ must satisfy the following transcendental equation:

$$0 = -3 + \gamma\ [\cosh(\sinh^{-1}(\frac{5}{\gamma}) - 1)].$$

The `ROOT` operator finds that value of a variable within a specified interval of values, where a given function of that variable equals zero. The form of the `ROOT` operator is `ROOT(X,A,B,E)` where `X` is the name of the unknown variable, `A` and `B` are numbers or expressions that specify the range of values within which the root-value lies, and `E` is a mathematical expression involving the unknown variable. The equation given above involving γ can be solved and the other unknowns evaluated, with the following MLAB commands:

```
FCT F() = ROOT(G,.001,100,-(3+G)+G*COSH(ASINH(5/G)))
A = 0; C = F(); XB = 2*C*ASINH(5/C); B = -(C+3);
TYPE A,B,C,XB
```

The ROOT operator finds a zero of the given mathematical expression by using a hybrid algorithm that mixes bracketing, secant, Newton, and bisection methods. These methods are described separately in Reference 2. COSH and ASINH evaluate the hyperbolic cosine and inverse hyperbolic sine, respectively. The TYPE command prints the following results to the screen:

```
* A = 0
* B = -5.66666667
* C = 2.66666667
* XB = 7.39356993
```

Another method of solving the four equations consists of using the MLAB MINIMIZE operator to find the values of α, β, γ, and $\bar{x}$ that make a least-squares objective function evaluate to the minimum value of 0. The least-squares objective function in this problem is given by:

$$\text{obj}() = [10 - L(-\frac{\bar{x}}{2}, \frac{\bar{x}}{2})]^2 + [y(\frac{\bar{x}}{2})]^2 + [y(-\frac{\bar{x}}{2})]^2 + [3 + y(0)]^2.$$

Here are the MLAB commands that use the MINIMIZE operator to solve the problem:

```
FCT Y(X) = B+C*COSH((X/C)+A)
FCT L(X0,X1) = C*(SINH((X1/C)+A)-SINH((X0/C)+A))
FCT OBJ() = (10-L(-XB/2,XB/2))^2+(Y(XB/2))^2+(Y(-XB/2))^2+(3+Y(0))^2
A = 1; C = 1; XB = 1; B = 1
MINIMIZE(OBJ,XB,A,B,C)
TYPE A,B,C,XB
```

With the first 3 commands, we define Y(X) to be the general form of the catenary function, L(X0,X1) to be the distance between the endpoints along the catenary, and OBJ() to be the least-squares objective function for the system of equations. The assignment statements provide initial guesses for the parameters A, B, C, and XB that appear in the definitions of the the functions. Then the MINIMIZE operation does a variable metric method search (as described in Reference 2) of the (α,β,γ,$\bar{x}$)-parameter space from the point (1,1,1,1). The first argument to MINIMIZE is the name of the function for which a minimum is to be found; the remaining arguments are the names of the parameters to be varied. The MINIMIZE operator changes the values of the parameters listed in the argument list. The results printed to the screen by the last TYPE command are:

```
* A = 8.86949601E-10
* B = -5.66666669
* C = 2.66666669
* XB = 7.39356995
```

Both the ROOT and MINIMIZE methods determined the separation of the pier and the boat to be approximately 7.39 feet.

The MLAB FIT and MINIMIZE operators can now be used to investigate how the minimum of the catenary changes as the right endpoint moves down or up vertically, i.e. as the tide ebbs or flows. Although the minimum of the catenary formed when the boat is above or below the pier is not the same as the minimum when the boat and the pier are at the same level, we continue to use the coordinate system described above in which the y-axis passes through the minimum of the catenary formed when the boat and the pier *are* at the same level. Note that the boat is always three feet above the water level.

```
/* compute the maximum height that the boat can attain if the rope
   is 10 feet long and the pier and boat are separated by xb feet */
MH = SQRT(100-XB^2);

/* compute a vector of 21 equally-spaced height values ranging from
   -MH to MH, excluding MH and -MH */
D = -MH:MH!23
DELETE D ROW (1,23)

/* L(-XB/2,XB/2) = 10, and ML = the corresponding data point */
ML = LIST(-XB/2,XB/2,10)'

/* define linear constraints for XMIN, B, C */
CONSTRAINTS Q1 = {B<0,C>0}
CONSTRAINTS Q2 = {XMIN >= -XB/2, XMIN <= XB/2}

XMIN = 0; /* initial guess at minimum's abscissa */
```

```
/* loop over 21 different heights, D[J], J = 1,2,...,21 */
FOR J = 1:21 DO {
    /* MC = 2X2 matrix containing endpoints of the catenary
       when the boat is at (XB/2,D[J]) */
    MC = SHAPE(2,2,LIST(-XB/2,0,XB/2,D[J]));

    /* adjust (A,B,C) to fit the catenary to the endpoints and length */
    FIT (A,B,C), Y TO MC, L TO ML, CONSTRAINTS Q1;

    /* find the minimum point (XMIN,YMIN) on the catenary, leaving
       A,B, and C fixed, constraining (-XB/2)<=XMIN<=(XB/2) */
    YMIN = MINIMIZE(Y,XMIN,Q2);

    /* store the results in the matrix MRES */
    MRES ROW J = LIST(D[J],XMIN,YMIN,A,B,C)';
   }
```

The MLAB `FIT` operator uses the Marquardt-Levenberg curve-fitting method described in References 3 and 4. The quadratic programming algorithm, which the curve-fitting algorithm needs to converge within linear constraints is described in Reference 5.

Five of the resulting catenaries are graphed via MLAB in Figure 2.



Figure 2: Catenaries for 5 different tide levels. The minimum of each catenary is marked with $\times$ and connected to a horizontal dash mark which represents the water level.

The computed minimum of each of the 21 catenaries and the corresponding water levels are shown in Figure 3.



Figure 3: The minima of 21 catenaries marked with × connected to horizontal dash marks which represent water level.

If the tide ebbs, thereby lowering the boat with respect to the pier, the rope clears the water. If the tide rises, thereby raising the boat with respect to the pier, the rope is partially submerged.

This paper has demonstrated the `ROOT`, `MINIMIZE`, and `FIT` operators in

MLAB within the context of a catenary problem. MLAB can be used for studying many other problems in physics and engineering.

For more information about MLAB, please contact:

Civilized Software Inc.
12109 Heritage Park Circle
Silver Spring, MD 20906 USA
phone: (301) 962-3711
email: csi@civilized.com

and visit our web page at http://www.civilized.com.

References:

1. Keith R. Symon, *Mechanics*. 3rd edition. (Reading, MA: Addison-Wesley Publishing Co., 1971). pages 239-241.

2. William H. Press, Saul A. Teukolsky, William T. Vetterling, Brian P. Flannery. *Numerical Recipes in FORTRAN: The Art of Scientific Computing*. 2nd edition. (Cambridge, England: Cambridge University Press, 1992). pages 340-360, 418-422.

3. D. W. Marquardt. *"An Algorithm for Least-Squares Estimation of Non-Linear Parameters"*. J. SIAM, vol 11, no. 2, pp. 431:441, 1963.

4. Lyle B. Smith. *"The Use of Interactive Graphics to Solve Numerical Problems"*. CACM, vol. 13, no. 10, pp. 625:634, October 1970.

5. Richard I. Shrager. *"Quadratic Programming for Non-Linear Regression"*. CACM, vol.15, no. 1, pp. 41:44. January 1972.